


NEW

**the
TEXAS
PROGRAM
BOOK**

**VINCE
APPS**

**35 PROGRAMS FOR GAMES HOME
& BUSINESS USE WITH THE TI 99/4A**

THE TEXAS PROGRAM BOOK



THE TEXAS PROGRAM BOOK

**35 programs for home, educational and
business use with the TI 99/4A**

Vince Apps

**Phoenix Publishing Associates
Bushey, Herts**

First published 1983 by
Phoenix Publishing Associates,
14 Vernon Road, Bushey, Herts. WD2 2JL
Reprinted 1983

ISBN 0 9465 7600 9 (paper)
© Vince Apps 1983

Filmset by Northumberland Press Ltd,
Gateshead, Tyne and Wear

Printed in Great Britain by
Richard Clay (The Chaucer Press) Ltd,
Bungay, Suffolk

Contents

FOREWORD	7
CHARACTER SET	9
Evasion	11
Invader.	16
3-D Maze	19
Caterpillar	25
Snake	28
Simon	32
Organ	34
Horse Race	36
Tennis	39
Space Attack	43
Towers of Hanoi	47
Anagram	50
Morse Code	53
Penguin	55
Lunar Lander	59
Space Zap	61
Kaleidoscope	64
Chase	65
Digital Clock	68
Binary to Hexadecimal Conversion	70
Hexadecimal to Decimal Conversion	72
Binary to Decimal Conversion.	73
Arithmetic	75
Spelling Test	78
Multiplication Tables	80
Report	81
Planet Lander	83
Entrapment	87
Print At	90
Cheque Book	91
Metric Converter	92
Alien Attack	94
Capitals	98
Shoebox	100
Moneybags	103

Foreword

If you are already the owner of a Texas TI 99/4A you will have a good idea of the range of sound, colour and graphics your machine possesses.

You may not yet, however, be able to write your own programs for pleasure, education or business use so we have supplied you, in this first book, with a range of programs for you to input to your machine. Some programs are short, and simple, to begin with and some will lead you towards writing your own programs in the future.

All are designed for use with Texas Basic.

We hope that these varied programs will give you hours of enjoyment from your home computer.

CHARACTER SET:

abcdefghijklmnopqrstuvwxyz

ABCDEFGHIJKLMNOPQRSTUVWXYZ

1234567890

!@#\$%^&*()+=-/,;X.

CHARACTERS TO NOTE:

J SEMI COLON

I COLON

J COMMA

Ø ZERO

O LETTER O

. FULL STOP



Evasion

Evasion has you being chased around a maze by 4 evil looking green ghosts intent on your downfall.

You can move around the maze by pressing the arrow keys (E S D X). Points are gained by just staying alive!

You will need to keep your wits about you as well as moving with lightning reactions, as this game needs some skill to play. 1500+ can be considered a respectable score.

```
100 REM
110 REM    EVASION
120 REM
130 GOTO 610
140 FOR J=1 TO LEN(STR$(SC))
150 CHR=ASC(SEG$(STR$(SC),J,1))
160 CALL HCHAR(1,19+J,CHR)
170 NEXT J
180 RETURN
190 REM    WHICH WAY?
200 CALL KEY(0,K,S)
210 MRN=MR0
220 MCN=MC0
230 IF K<>63 THEN 250
240 MCN=MCN-1
250 IF K<>68 THEN 270
260 MCN=MCN+1
270 IF K<>69 THEN 290
280 MRN=MRN-1
290 IF K<>68 THEN 310
300 MRN=MRN+1
```

```

310 IF (MCN<3)+(MCN>32)+(MRN<1)+(MRN>24)
    THEN 410
320 CALL GCHAR(MRN,MCN,X)
330 IF X=140 THEN 1210
340 IF (X=120)+(X=121)+(X=122) THEN 410
350 IF (MRN=MRO)*(MCN=MCO) THEN 410
360 CALL HCHAR(MRO,MCO,32)
370 CALL SOUND(100,330,2)
380 CALL HCHAR(MRN,MCN,130)
390 MRO=MRN
400 MCO=MCN
410 RETURN
420 REM    MOVE GHOSTS
430 R=INT(RND*4)+1
440 IF NS(R,1)=MRN THEN 510
450 CALL HCHAR(NS(R,1),NS(R,2),STR(R))
460 NS(R,1)=NS(R,1)+SGN(MRN-NS(R,1))
470 CALL GCHAR(NS(R,1),NS(R,2),STR(R))
480 IF STR(R)<>140 THEN 500
490 STR(R)=32
500 CALL HCHAR(NS(R,1),NS(R,2),140)
505 CALL SOUND(100,220,2)
510 IF NS(R,2)=MCN THEN 590
520 CALL HCHAR(NS(R,1),NS(R,2),STR(R))
530 NS(R,2)=NS(R,2)+SGN(MCN-NS(R,2))
540 CALL GCHAR(NS(R,1),NS(R,2),STR(R))
550 IF STR(R)<>140 THEN 580
560 STR(R)=32
580 CALL HCHAR(NS(R,1),NS(R,2),140)
590 IF (MRN=NS(R,1))*(MCN=NS(R,2))
    THEN 1210
600 RETURN
610 CALL CLEAR

```

```

620 RANDOMIZE
630 CALL SCREEN(8)
640 CALL CHAR(120,"000000FFFF")
650 REM   SET UP MAZE
660 FOR J=8 TO 20 STEP 6
670 CALL HCHAR(J,5,120,8)
680 CALL HCHAR(J,14,120,9)
690 CALL HCHAR(J,24,120,7)
700 NEXT J
710 FOR J=4 TO 22 STEP 6
720 CALL HCHAR(J,5,120,13)
730 CALL HCHAR(J,19,120,12)
740 NEXT J
750 FOR J=6 TO 18 STEP 6
760 CALL HCHAR(J,5,120,4)
770 CALL HCHAR(J,10,120,16)
780 CALL HCHAR(J,27,120,4)
790 NEXT J
800 CALL HCHAR(24,5,120,26)
810 CALL HCHAR(2,5,120,26)
820 CALL CHAR(121,"0101010101010101")
830 CALL VCHAR(2,4,121,23)
840 CALL CHAR(122,"8080808080808080")
850 CALL VCHAR(2,31,122,23)
860 CALL CHAR(130,"3844386CAA3844C6")
870 CALL COLOR(13,5,1)
880 CALL CHAR(140,"387CBAFFFFC63838")
890 CALL COLOR(14,3,1)
900 REM   SET INITIAL POSITIONS
910 NS(1,1)=3
920 NS(1,2)=5
930 NS(2,1)=3
940 NS(2,2)=30

```

```

950 NS(3,1)=23
960 NS(3,2)=5
970 NS(4,1)=23
980 NS(4,2)=30
990 M$="SCORE = "
1000 PR=1
1010 PC=11
1020 GOSUB 1330
1030 FOR J=1 TO 4
1040 STR(J)=32
1050 NEXT J
1060 FOR J=1 TO 4
1070 CALL HCHAR(NS(J,1),NS(J,2),140)
1080 NEXT J
1090 MRN=11
1100 MRO=11
1110 MCN=18
1120 MCO=18
1130 CALL HCHAR(MRN,MCN,130)
1140 GOSUB 200
1150 GOSUB 430
1160 GOSUB 200
1170 SC=SC+10
1180 GOSUB 140
1190 GOTO 1140
1200 REM   CAPTURED
1210 FOR J=660 TO 110 STEP -20
1220 CALL SOUND(200,J,2)
1230 NEXT J
1240 M$="PLAY AGAIN?"
1250 PR=13
1260 PC=11
1270 GOSUB 1330

```

```
1280 CALL KEY(0,K,S)
1290 IF S=0 THEN 1280
1300 IF K<>89 THEN 1380
1310 SC=0
1320 GOTO 130
1330 FOR J=1 TO LEN(M$)
1340 CHR=ASC(SEG$(M$,J,1))
1350 CALL HCHAR(PR,PC+J,CHR)
1360 NEXT J
1370 RETURN
1380 END
```


Invader

It's time to defend the earth as you shoot down a never-ending stream of alien invaders. Allow three to escape and the game ends.

Use the space bar to fire.

```
100 REM
110 REM    INVADER
120 REM
130 CALL CLEAR
140 RANDOMIZE
150 CN=3
160 CALL CLEAR
170 CALL SCREEN(15)
180 REM    DEFINE CHARACTERS
190 CALL COLOR(13,13,1)
200 CALL CHAR(130,"3C7EBDBDFF3C4281")
210 CALL COLOR(14,5,1)
220 CALL CHAR(136,"1818FFFFFFFFFFFFFF")
230 CALL COLOR(15,16,1)
240 CALL CHAR(144,"0608080808080808")
250 REM    INVADER POSITION
260 R=INT(RND*21+2)
270 F=0
280 AP=1
290 GOSUB 740
300 CALL HCHAR(24,15,136)
310 AP=AP+1
320 CALL HCHAR(R,AP-1,32)
330 CALL HCHAR(R,AP,130)
340 IF AP<30 THEN 400
```

```

350 CALL CLEAR
360 CN=CN-1
370 IF CN=0 THEN 570
380 GOSUB 740
390 GOTO 260
400 IF F=1 THEN 560
410 REM   FIRE?
420 CALL KEY(0,K,S)
430 IF K<>32 THEN 560
440 F=1
450 CALL SOUND(200,-6,2)
460 CALL VCHAR(1,15,144,23)
470 CALL VCHAR(1,15,32,23)
480 IF AP<>15 THEN 560
490 CALL SOUND(750,-7,2)
500 CALL SCREEN(9)
510 SC=SC+100
520 GOSUB 670
530 CALL SCREEN(15)
540 GOSUB 740
550 GOTO 260
560 GOTO 300
570 PRINT "YOUR SCORE:";SC;:
580 INPUT "ANOTHER GAME? ":Q$
590 IF (SEG$(Q$,1,1)="N")+ (SEG$(Q$,1,1)
    ="n") THEN 650
600 SC=0
610 CN=0
620 CN=3
630 CALL CLEAR
640 GOTO 260
650 STOP
660 REM   SUBROUTINES

```

```

670 FOR D=1 TO 100
680 NEXT D
690 RETURN
700 FOR MS=1 TO LEN(M#)
710 CALL HCHAR(1,PSN+MS,ASC(SEG$(M#,
    MS,1)))
720 NEXT MS
730 RETURN
740 PSN=5
750 M#="BASES = "&STR$(CN)
760 GOSUB 700
770 PSN=17
780 M#="SCORE = "&STR$(SC)
790 GOSUB 700
800 RETURN

```

3-D Maze

This program gives an excellent demonstration of the graphics capabilities of the TI 99/4A. You see a perspective view of the inside of a maze as you wander through it, trying to find the way out!

To face in a compass direction, press keys N, S, E, or W. To move forwards press F. Note that you can only move by pressing F, although the direction keys will change the viewpoint.

```
100 REM
110 REM      3-D MAZE
120 REM
130 CALL CLEAR
140 CALL SCREEN(16)
150 REM      DEFINE CHARACTERS
160 CALL COLOR(13,3,3)
170 CALL COLOR(14,3,1)
180 CALL CHAR(130,"FFFFFFFFFFFFFFFF")
190 CALL CHAR(138,"0103070F1F3F7FFF")
200 CALL CHAR(139,"FEFCF8F0E0C080")
210 CALL CHAR(140,"80C0E0F0F8FCFEFF")
220 CALL CHAR(141,"7F3F1F0F070301")
230 XV=10
240 YV=13
250 PRINT "3-D MAZE":
260 INPUT "skill level? type a number
        between 5 (easy) and 90
        (difficult) and press enter." :LN
270 IF (LN>4)*(LN<91)THEN 300
280 PRINT : "5 TO 90 PLEASE":
290 GOTO 260
```

```

300 DIM M(90,12)
310 PRINT "wait..."
320 RANDOMIZE
330 GOTO 760
340 REM   END WALL
350 X=XV
360 Y=YV
370 FOR I=1 TO 5
380 CALL VCHAR(X,Y,130,6)
390 Y=Y+1
400 NEXT I
410 RETURN
420 REM   SIDE WALL
430 X=XV
440 Y=YV+5
450 N=6
460 FOR I=1 TO 9
470 CALL VCHAR(X,Y,130,N)
480 CALL VCHAR(X-1,Y,138)
490 CALL VCHAR(X+N,Y,141)
500 N=N+2
510 X=X-1
520 Y=Y+1
530 NEXT I
540 RETURN
550 REM   SIDE WALL
560 X=XV
570 Y=YV
580 N=6
590 FOR I=1 TO 9
600 CALL VCHAR(X,Y,130,N)
610 CALL VCHAR(X+N,Y,139)
620 CALL VCHAR(X-1,Y,140)

```

```

630 X=X-1
640 Y=Y-1
650 N=N+2
660 NEXT I
670 RETURN
680 REM SIDE WALL DOOR
690 GOSUB 420
700 CALL VCHAR(XV-6,YV+10,32,44)
710 RETURN
720 REM SIDE WALL DOOR
730 GOSUB 550
740 CALL VCHAR(XV-7,YV-6,32,44)
750 RETURN
760 GOSUB 1590
770 FOR I=1 TO LN
780 FOR J=1 TO 12
790 R=INT(RND*1.9+1)
800 IF M(I,J)<>0 THEN 820
810 M(I,J)=R
820 NEXT J
830 NEXT I
840 FOR I=1 TO 12
850 M(LN,I)=1
860 M(LN-1,I)=2
870 M(1,I)=4
880 IF M(2,I)=3 THEN 900
890 M(2,I)=1
900 NEXT I
910 FOR I=1 TO LN
920 M(I,1)=1
930 M(I,12)=1
940 NEXT I
950 R=LN-1

```

```

960 C=2
970 CALL CLEAR
980 ON D+1 GOTO 990,1070,1030,1110
990 ON M(R-1,C)GOSUB 340,410,410,410
1000 ON M(R,C-1)GOSUB 550,720,720,720
1010 ON M(R,C+1)GOSUB 420,680,680,680
1020 GOTO 1140
1030 ON M(R+1,C)GOSUB 340,410,410,410
1040 ON M(R,C+1)GOSUB 550,720,720,720
1050 ON M(R,C-1)GOSUB 420,680,680,680
1060 GOTO 1140
1070 ON M(R,C+1)GOSUB 340,410,410,410
1080 ON M(R-1,C)GOSUB 550,720,720,720
1090 ON M(R+1,C)GOSUB 420,680,680,680
1100 GOTO 1140
1110 ON M(R,C-1)GOSUB 340,410,410,410
1120 ON M(R+1,C)GOSUB 550,720,720,720
1130 ON M(R-1,C)GOSUB 420,680,680,680
1140 REM PRINT DIRECTION
1150 IF D<>0 THEN 1170
1160 CALL HCHAR(23,28,78)
1170 IF D<>2 THEN 1190
1180 CALL HCHAR(23,28,83)
1190 IF D<>1 THEN 1210
1200 CALL HCHAR(23,28,69)
1210 IF D<>3 THEN 1240
1220 CALL HCHAR(23,28,87)
1230 REM WHICH WAY?
1240 CALL KEY(0,K,S)
1250 IF S=0 THEN 1240
1260 IF K=70 THEN 1320
1270 IF K=78 THEN 1470
1280 IF K=83 THEN 1490

```

```

1290 IF K=69 THEN 1510
1300 IF K=87 THEN 1530
1310 GOTO 1240
1320 MV=MV+1
1330 ON D+1 GOTO 1340,1410,1380,1440
1340 IF M(R-1,C)=1 THEN 1550
1350 R=R-1
1360 IF M(R,C)=4 THEN 1560
1370 GOTO 970
1380 IF M(R+1,C)=1 THEN 1550
1390 R=R+1
1400 GOTO 970
1410 IF M(R,C+1)=1 THEN 1550
1420 C=C+1
1430 GOTO 970
1440 IF M(R,C-1)=1 THEN 1550
1450 C=C-1
1460 GOTO 970
1470 D=0
1480 GOTO 970
1490 D=2
1500 GOTO 970
1510 D=1
1520 GOTO 970
1530 D=3
1540 GOTO 970
1550 GOTO 1140
1560 PRINT "congratulations! you
      escaPedin only";MV;"moves":;:
1570 STOP
1580 REM    PLOT MAZE
1590 R4=1
1600 R1=INT(RND*7+3)

```



```

1610 R2=INT(RND*4+2)
1620 FOR I=R4 TO R4+R2
1630 IF I=LN-1 THEN 1810
1640 M(I,R1)=3
1650 NEXT I
1660 R4=R4+R2
1670 M(R4+1,R1)=1
1680 R3=INT(RND*2+1)
1690 IF R3<>2 THEN 1710
1700 R3=-1
1710 R5=INT(RND*3+2)
1720 FOR I=1 TO R5
1730 IF (R1+R3*I)<11 THEN 1750
1740 R1=R1-1
1750 IF (R1+R3*I)>1 THEN 1770
1760 R1=R1+1
1770 M(R4,R1+R3*I)=3
1780 NEXT I
1790 R1=R1+R5*R3
1800 GOTO 1610
1810 RETURN

```

Caterpillar

You are in control of the body of a fast growing caterpillar. Scattered around the screen are spiders which must be avoided. If you hit either a spider or your own body the game ends.

The control keys are: 1 - Up
2 - Down
9 - Left
0 - Right

One point is scored for each segment of body produced.

```
100 REM
110 REM    CATERPILLAR
120 REM
130 CALL CLEAR
140 RANDOMIZE
150 PRINT "CATERPILLAR": : : :
160 PRINT "control keys": : :
170 PRINT "1 = up"
180 PRINT "2 = down"
190 PRINT "9 = left"
200 PRINT "0 = right": :
210 INPUT "Press enter to start": NUL$
220 DIM MEM(32,32)
230 CALL COLOR(14,2,1)
240 CALL CHAR(130,"183C7EFFFF7E3C18")
250 CALL CHAR(136,"8124997E187E9924")
260 X=4
270 Y=4
280 K=48
290 CALL SCREEN(16)
300 CALL CLEAR
```

```

310 REM   SET UP SCREEN
320 FOR J=1 TO 29
330 R1=INT(RND*24+4)
340 R2=INT(RND*23+1)
350 MEM(R2,R1)=3
360 CALL HCHAR(R2,R1,136)
370 NEXT J
380 CALL COLOR(13,13,1)
390 REM   PRINT BODY
400 IF X<1 THEN 660
410 IF Y<4 THEN 680
420 IF X>24 THEN 700
430 IF Y>28 THEN 720
440 CALL SOUND(50,262,7)
450 CALL HCHAR(X,Y,130)
460 SC=SC+1
470 IF MEM(X,Y)=3 THEN 760
480 MEM(X,Y)=3
490 REM   WHICH WAY?
500 CALL KEY(0,KY,S)
510 IF S=0 THEN 530
520 K=KY
530 IF K=50 THEN 580
540 IF K=48 THEN 600
550 IF K=49 THEN 620
560 IF K=57 THEN 640
570 GOTO 400
580 X=X+1
590 GOTO 400
600 Y=Y+1
610 GOTO 400
620 X=X-1
630 GOTO 400

```

```

640 Y=Y-1
650 GOTO 400
660 X=1
670 GOTO 730
680 Y=4
690 GOTO 730
700 X=24
710 GOTO 730
720 Y=28
730 MEM(X,Y)=0
740 SC=SC-1
750 GOTO 400
760 CALL SOUND(1000,-5,2)
770 PRINT "your score is",SC:;
780 INPUT "play again? (Y/N)":Q$
790 IF (SEG$(Q$,1,1)="n")+ (SEG$(Q$,1,1)
    ="N") THEN 870
800 FOR J=1 TO 32
810 FOR K=1 TO 32
820 MEM(J,K)=0
830 NEXT K
840 NEXT J
850 SC=0
860 GOTO 220
870 END

```

Snake

This could be called "Caterpillar's Revenge"!

It's the spiders turn to be eaten as you move a fast growing snake around the screen. Hazards in this game are a number of randomly placed rocks.

Control keys are: E - Up
X - Down
S - Left
D - Right

Points are awarded for the number of spiders eaten. Eat enough and a new screenful will be generated!

```
100 REM
110 REM    SNAKE
120 REM
130 CALL CLEAR
140 RANDOMIZE
150 PRINT "SNAKE":...
160 PRINT "control keys":...
170 PRINT "E = up"
180 PRINT "X = down"
190 PRINT "S = left"
200 PRINT "D = right":...
210 INPUT "Press enter to start":NUL$
220 DIM MEM(32,32)
230 REM    DEFINE CHARACTERS
240 CALL COLOR(14,15,1)
250 CALL CHAR(130,"163C7EFFFF7E3C18")
260 CALL CHAR(136,"8124997E187E9924")
270 CALL COLOR(15,4,1)
```

```

280 CALL CHAR(144,"387EFFFFFFFF7E")
290 X=4
300 Y=4
310 K=68          (6)
320 CALL SCREEN(2)
330 CALL CLEAR
340 REM   POSITION ROCKS & SPIDERS
350 FOR J=1 TO 29
360 R1=INT(RND*24+4)
370 R2=INT(RND*23+1)
380 IF MEM(R2,R1)=2 THEN 360
390 MEM(R2,R1)=2
400 CALL HCHAR(R2,R1,136)
410 NEXT J
420 FOR J=1 TO 20
430 R1=INT(RND*24+4)
440 R2=INT(RND*23+1)
450 MEM(R2,R1)=3
460 CALL HCHAR(R2,R1,144)
470 NEXT J
480 CALL COLOR(13,7,1)
490 IF X<1 THEN 770
500 IF Y<4 THEN 790
510 IF X>24 THEN 810
520 IF Y>28 THEN 830
530 CALL SOUND(50,262,7)
540 CALL HCHAR(X,Y,130)
550 IF MEM(X,Y)<>2 THEN 580
560 SC=SC+10
570 IF SC/280=INT(SC/280) THEN 920
580 IF MEM(X,Y)=3 THEN 860
590 MEM(X,Y)=3
600 REM   WHICH WAY?

```

```

610 CALL KEY(0,KY,S)
620 IF S=0 THEN 640
630 K=KY
640 IF K=88 THEN 690
650 IF K=68 THEN 710
660 IF K=69 THEN 730
670 IF K=83 THEN 750
680 GOTO 490
690 X=X+1
700 GOTO 490
710 Y=Y+1
720 GOTO 490
730 X=X-1
740 GOTO 490
750 Y=Y-1
760 GOTO 490
770 X=1
780 GOTO 840
790 Y=4
800 GOTO 840
810 X=24
820 GOTO 840
830 Y=28
840 MEM(X,Y)=0
850 GOTO 490
860 CALL SOUND(1000,-5,2)
870 CALL SCREEN(16)
880 PRINT "your score is",SC:,,,
890 INPUT "Play again? (Y/N)":Q$
900 IF (SEG$(Q$,1,1)="n")+ (SEG$(Q$,1,1)
    ="N") THEN 980
910 SC=0
920 FOR J=1 TO 32

```

```
930 FOR K=1 TO 32
940 MEM(J,K)=0
950 NEXT K
960 NEXT J
970 GOTO 220
980 END
```


Simon

This is the traditional game of Simon.

The computer will print a sequence of numbers for you to memorise.

You will be asked to repeat the sequence. If you guess correctly, the length of the sequence will be increased for your next attempt.

```
100 REM
110 REM    SIMON
120 REM
130 CALL CLEAR
140 CN=1
150 PRINT "your sequence:";:
160 R$=""
170 FOR I=1 TO CN
180 RANDOMIZE
190 R=INT(RND*9+1)
200 PRINT R;
210 IF I/9<>INT(I/9)THEN 230
220 PRINT :;
230 R$=R$&STR$(R)
240 NEXT I
250 FOR DEL=1 TO 700
260 NEXT DEL
270 CALL CLEAR
280 INPUT "your guess:";G$
290 IF G$=R$ THEN 320
300 PRINT "the correct sequence was:";R$
310 GOTO 350
320 PRINT "correct!";:
```

```

330 CN=CN+1
340 GOTO 350
350 INPUT "again? ":Q#
360 IF (SEG$(Q#,1,1)<>"n")*(SEG$(Q#,1,
1)<>"N")THEN 150
370 PRINT "you correctly remembered a
sequence of";CN-1;"numbers."
380 END

```

Organ

This program turns your TI 99 into a mini electronic organ! The top letter row (starting with Q) forms the keyboard.

As presented it covers 11 notes around middle C but this can be easily extended to use the entire keyboard by the extension of the data statements in lines 370 onwards.

After playing a sequence of notes, press the space bar for a repeat performance!

```
100 REM
110 REM   ORGAN
120 REM
130 CALL CLEAR
140 DIM LTR(11),NOTE(11),MEM(100)
150 REM   READ IN DATA
160 FOR J=1 TO 11
170 READ LTR(J)
180 READ NOTE(J)
190 NEXT J
200 PRINT "READY":
210 REM   LOOK AT KEYBOARD
220 CALL KEY(0,K,S)
230 IF K=32 THEN 330
240 FOR J=1 TO 11
250 IF LTR(J)=K THEN 260
260 NEXT J
270 GOTO 220
280 CNT=CNT+1
290 MEM(CNT)=NOTE(J)
300 CALL SOUND(500,NOTE(J),2)
```

```
310 GOTO 220
320 REM   PLAY MEMORY
330 FOR J=1 TO CNT
340 CALL SOUND(500, MEM(J), 2)
350 NEXT J
360 GOTO 220
370 DATA 81,165,87,175,69,196,82,220
380 DATA 84,247,89,262,85,294,73,330
390 DATA 79,349,80,392,47,440
```

Horse Race

A horse race simulation for penniless punters. You start with £100, and bet on any one of five horses. Surprisingly addictive, even for non-gamblers.

```
100 REM
110 REM    HORSE RACE
120 REM
130 CALL CLEAR
140 CALL SCREEN(16)
150 RANDOMIZE
160 DEF RFN=INT(RND*5+1)
170 SUM=100
180 REM    ASSIGN STRING VALUES
190 A$(1)="born loser"
200 A$(2)="golden wonder"
210 A$(3)="blue rum"
220 A$(4)="space invader"
230 A$(5)="Punter's despair"
240 PRINT "HORSE RACE":
250 PRINT "you are betting on the
    outcome of a horse race.":
260 PRINT "you start with 100 Pounds.
    good luck!":
270 REM    MAIN LOOP
280 PRINT "running are:":
290 PRINT " 1. born loser"
300 PRINT " 2. golden wonder"
310 PRINT " 3. blue rum"
320 PRINT " 4. space invader"
330 PRINT " 5. Punter's despair":
340 PRINT "all horses pay 3-1.":
```

```

350 INPUT "how much do you bet? " : BET
360 IF (BET>SUM)+(BET<0) THEN 350
370 PRINT " "
380 INPUT "on what number horse? " : HRS
390 PRINT " "
400 IF (HRS<1)+(HRS>5) THEN 380
410 PRINT "they're off!" : " "
420 GOSUB 710
430 R=RFN
440 PRINT "it's ";A$(R); " in the
    lead." : " "
450 GOSUB 710
460 R1=RFN
470 IF R1=R THEN 460
480 PRINT "running neck and neck is
    ";A$(R1); " ." : " "
490 GOSUB 710
500 PRINT "it's ";A$(R1); " in the
    lead." : " "
510 GOSUB 710
520 R2=RFN
530 IF (R2=R1)+(R2=R) THEN 520
540 PRINT A$(R2); " is pulling ahead." : " "
550 GOSUB 710
560 PRINT "and it's ";A$(R2); " the
    winner!" : " "
570 IF HRS<>R2 THEN 630
580 SUM=SUM+3*BET
590 PRINT "you win! you now have";SUM;
    "Pounds." : " "
600 GOSUB 710
610 GOSUB 710
620 GOTO 670

```

```
630 SUM=SUM-BET
640 IF SUM<=0 THEN 680
650 PRINT "you lose! you now have";SUM;
    "Pounds."::
660 GOTO 600
670 GOTO 280
680 PRINT "you're broke!"::
690 STOP
700 REM    DELAY
710 FOR K=1 TO 1000
720 NEXT K
730 RETURN
```

Tennis

For two players. Each player controls a bat, player one uses keys Q and A for the left hand bat, player two uses keys O and L for the right hand bat.

A player's score is incremented by one if the opposing player misses the ball. The game ends when either players score reaches 9.

This program uses routines that can form the basis of many bat and ball games. Note particularly the move ball subroutine and the two bat subroutines.

Budding programmers might like to extend the game to give a one player option by replacing either player subroutine with one giving the computer control of the bat.

```
100 REM
110 REM    TENNIS
120 REM
130 GOTO 770
140 REM    MOVE BALL
150 CALL HCHAR(YP,XP,32)
160 XP=XP+XD
170 YP=YP+YD
180 IF (YP>22)+(YP<4)THEN 190 ELSE 210
190 YD=-YD
200 CALL SOUND(100,200,2)
210 CALL GCHAR(YP,XP,CH)
220 IF CH<>130 THEN 290
230 CALL SOUND(100,300,2)
240 XD=-XD
250 CALL VCHAR(YP,XP,130)
```



```

260 YP=YP+YD
270 XP=XP+XD
280 GOTO 310
290 IF (XP<2)+(XP>31)THEN 600
300 CALL HCHAR(YP,XP,140)
310 RETURN
320 REM    PLAYER ONE
330 CALL KEY(0,K,S)
340 IF K<>61 THEN 360
350 R1=R1-2
360 IF K<>65 THEN 380
370 R1=R1+2
380 IF R1=R10 THEN 450
390 IF (R1>2)*(R1<24)THEN 420
400 R1=R10
410 GOTO 450
420 CALL VCHAR(R10,C1,32,2)
430 CALL VCHAR(R1,C1,130,2)
440 R10=R1
450 RETURN
460 REM    PLAYER TWO
470 CALL KEY(0,K,S)
480 IF K<>79 THEN 500
490 R2=R2-2
500 IF K<>76 THEN 520
510 R2=R2+2
520 IF R2=R20 THEN 590
530 IF (R2>2)*(R2<24)THEN 560
540 R2=R20
550 GOTO 590
560 CALL VCHAR(R20,C2,32,2)
570 CALL VCHAR(R2,C2,130,2)
580 R20=R2

```

```

590 RETURN
600 CALL HCHAR(YP,XP,32)
610 CALL SOUND(100,120,2)
620 IF XP>3 THEN 670
630 SC2=SC2+1
640 CALL HCHAR(1,22,ASC(STR$(SC2)))
650 IF SC2=9 THEN 740
660 GOTO 700
670 SC1=SC1+1
680 CALL HCHAR(1,11,ASC(STR$(SC1)))
690 IF SC1=9 THEN 740
700 XP=16
710 YP=12
720 YD=-YD
730 GOTO 300
740 FOR DEL=1 TO 1500
750 NEXT DEL
760 END
770 CALL CLEAR
780 R10=12
790 R20=12
800 R1=12
810 R2=12
820 C1=4
830 C2=31
840 CALL CHAR(130,"1818181818181818")
850 CALL CHAR(140,"000010367C3810")
860 CALL VCHAR(R1,C1,130,2)
870 CALL VCHAR(R2,C2,130,2)
880 XD=1
890 YD=1
900 XP=14
910 YP=12

```

```
920 REM    MAIN LOOP
930 GOSUB 150
940 GOSUB 330
950 GOSUB 150
960 GOSUB 470
970 GOTO 930
```

Space Attack

You are being attacked by hordes of aliens, which are appearing at random on the screen. Move your laser sight by using the arrow keys (E, S, D, X). Fire by pressing the space bar.

You will be overrun and destroyed if the number of aliens exceeds twelve. Since you are in E-space, if you fire into empty space (i.e. miss!), your laser will be reflected back and you will suffer an equally rapid demise! Remember this, since aliens have a habit of disappearing suddenly!

```
100 REM
110 REM    SPACE ATTACK
120 REM
130 CALL CLEAR
140 CALL SCREEN(16)
150 PRINT "SPACE ATTACK":...:
160 INPUT "skill level? (1=suicide,
    50=easy) ":SK
170 IF (SK>50)+(SK<1)THEN 160
180 CALL CLEAR
190 CALL SCREEN(2)
200 RANDOMIZE
210 R=12
220 C=16
230 REM    DEFINE CHARACTERS
240 CALL COLOR(1,16,1)
250 CALL COLOR(2,16,1)
260 REM
270 CALL CHAR(34,"00183C7E567E66")
280 CALL CHAR(33,"000018183C18")
290 CALL CHAR(40,"FF8181818181FF")
```

```

300 CALL CHAR(41,"FF819999BD9981FF")
310 CALL CHAR(42,"FF99BDFFD7FFE7FF")
320 DEF RFR=INT(RND*23+1)
330 DEF RFC=INT(RND*24+4)
340 REM    RANDOM POSITIONS
350 RW=RFR
360 CL=RFC
370 CALL GCHAR(RW,CL,W)
380 IF W<>32 THEN 350
390 RV=INT(RND*SK+1)
400 IF RV>2 THEN 430
410 CALL HCHAR(RW,CL,34)
420 ECN=ECN+1
430 IF ECN>12 THEN 490
440 REM    MOVE LASER SIGHT
450 CALL KEY(0,K,S)
460 IF K<>32 THEN 680
470 CALL GCHAR(R,C,W)
480 IF W<>40 THEN 600
490 CALL SOUND(1000,-6,2)
500 CALL SCREEN(9)
510 X=SIN(X)
520 PRINT "your score is";SC
530 CALL SCREEN(16)
540 INPUT "another game?":Q$
550 IF (SEG$(Q$,1,1)="N")+ (SEG$(Q$,1,1)
    ="n") THEN 590
560 SC=0
570 ECN=0
580 GOTO 180
590 STOP
600 SC=SC+10
610 ECN=ECN-1

```

```

620 CALL HCHAR(R,C,40)
630 CALL SOUND(500,110,2,-5,2)
640 CALL COLOR(2,9,1)
650 X=SIN(X)
660 CALL COLOR(2,16,1)
670 GOTO 350
680 IF K<>69 THEN 750
690 IF R=1 THEN 750
700 CALL HCHAR(R,C,32)
710 IF MQ>39 THEN 730
720 CALL HCHAR(R,C,MQ)
730 R=R-1
740 GOTO 950
750 IF K<>68 THEN 820
760 IF R=23 THEN 820
770 CALL HCHAR(R,C,32)
780 IF MQ>39 THEN 800
790 CALL HCHAR(R,C,MQ)
800 R=R+1
810 GOTO 950
820 IF K<>68 THEN 890
830 IF C=28 THEN 890
840 CALL HCHAR(R,C,32)
850 IF MQ>39 THEN 870
860 CALL HCHAR(R,C,MQ)
870 C=C+1
880 GOTO 950
890 IF K<>83 THEN 950
900 IF C=4 THEN 950
910 CALL HCHAR(R,C,32)
920 IF MQ>39 THEN 940
930 CALL HCHAR(R,C,MQ)
940 C=C-1

```

```
950 CALL GCHAR(R,C,W)
960 MO=W
970 W=W-31
980 ON W GOTO 990,1010,1030,990,1010,
    1030,990,990,990,1010,1030
990 CALL HCHAR(R,C,40)
1000 GOTO 350
1010 CALL HCHAR(R,C,41)
1020 GOTO 350
1030 RC=INT(RND*8+1)
1040 IF RC<>1 THEN 1070
1050 ECN=ECN-1
1060 GOTO 990
1070 CALL HCHAR(R,C,42)
1080 GOTO 350
```

Towers of Hanoi

This program simulates the well known puzzle, Towers of Hanoi.

Move the column numbers, one number at a time, from the left hand tower (no. 1) to the right hand tower (no. 3), in as few moves as possible. You may not place a larger number on top of a smaller number.

```
100 REM
110 REM    TOWERS OF HANOI
120 REM
130 CALL CLEAR
140 CN=1
150 INPUT "number of levels (1 to
    10)? " :ND
160 IF (ND>10)+(ND<1)THEN 630
170 FOR I=11-ND TO 10
180 A(1,I)=I
190 NEXT I
200 REM    PRINT TOWERS
210 FOR J=1 TO 10
220 FOR I=1 TO 3
230 IF A(I,J)=0 THEN 250
240 PRINT TAB(I*5);A(I,J);
250 NEXT I
260 PRINT
270 NEXT J
280 FOR I=11-ND TO 10
290 IF A(3,I)<>I THEN 330
300 NEXT I
310 PRINT "..."
320 GOTO 600
```



```

330 PRINT : : : :
340 NM=NM+1
350 INPUT "move from which tower? ":K
360 PRINT
370 REM  VALIDATE INPUT
380 IF K<1 THEN 350
390 IF K>3 THEN 350
400 FOR I=1 TO 10
410 IF A(K,I)=0 THEN 450
420 TH=A(K,I)
430 A(K,I)=0
440 GOTO 470
450 NEXT I
460 GOTO 350
470 INPUT "to where? ":K
480 PRINT : :
490 REM  VALIDATE INPUT
500 IF K<1 THEN 470
510 IF K>3 THEN 470
520 FOR I=1 TO 10
530 IF A(K,I)=0 THEN 570
540 IF A(K,I)<TH THEN 650
550 A(K,I-1)=TH
560 GOTO 210
570 NEXT I
580 A(K,10)=TH
590 GOTO 210
600 PRINT "congratulations! you
    completed it in only";
610 PRINT NM;"moves": : : :
620 STOP
630 PRINT "between 1 and 10 Please": :
640 GOTO 150

```

```
650 PRINT "you can't do that!":  
660 GOTO 470
```

Anagram

This program will generate a series of anagrams to be deciphered. If your guess is correct a new anagram will be given. Type "QUIT" to give up and "STOP" to end the game. The list of words in the data statements in lines 520-580 can of course be changed to suit individual preferences.

```
100 REM
110 REM   ANAGRAM
120 REM
130 CALL CLEAR
140 RANDOMIZE
150 DIM WD$(200),M$(12)
160 J=1
170 REM   READ IN DATA
180 READ WD$(J)
190 IF WD$(J)="END" THEN 220
200 J=J+1
210 GOTO 180
220 J=J-1
230 REM   SELECT WORD
240 R=INT(RND*J+1)
250 A$=WD$(R)
260 FOR I=1 TO LEN(A$)
270 R=INT(RND*12+1)
280 IF M$(R)<>"" THEN 270
290 M$(R)=SEG$(A$,I,1)
300 NEXT I
310 FOR I=1 TO 12
320 IF M$(I)=" " THEN 340
330 PRINT M$(I);
340 NEXT I
```

```

350 PRINT :
360 INPUT "your guess? ":G$
370 PRINT :
380 IF G$=A$ THEN 430
390 IF G$="QUIT" THEN 500
400 IF G$="STOP" THEN 600
410 PRINT "try again!":
420 GOTO 360
430 PRINT "correct!":
440 NC=NC+1
450 A$=""
460 FOR I=1 TO 12
470 M$(I)=" "
480 NEXT I
490 GOTO 240
500 PRINT A$:
510 GOTO 450
520 DATA POND,WOOD,MOUSE,TIGER,ANTELOPE,
    CASSETTE,SERENDIPITY,COMPUTER,
    PROGRAMME,APPLE,WORD,TRAIN,HOUSE,APEX
530 DATA DERELICT,DIGIT,BUTTERFLY,IDEA,
    INDEPENDENT,ANAGRAM,MINOTAUR,DUNGEON,
    MINIATURE,ENVIRONMENT,MOTH,PARTY
540 DATA OCEAN,ORBIT,OXYGEN,PASSPORT,
    PENNY,PENINSULA,RABBIT,PEAR,TOAD,
    POUND,GUESS,BINARY,PUPIL,BASIC,
    ANALOGUE
550 DATA VIDEO,BUG,RECORD,MAZE,SOFTWARE,
    MODULAR,HELICOPTER,TELEVISION,
    TELEPHONE,CONTROL,MACHINE,CABLE,
    ISSUE,LEAF
560 DATA DICE,CHESS,PROGRESS,SHARK,BOOK,
    ROBOT,PLAY,FILM,EVENT,GOOSE,FRESH,

```

```
JUNIOR, CRICKET, PICTURE, HEAP, HAVOC
570 DATA EXPERT, LANGUAGE, TRAVEL, VILLAGE,
    GENERATION, MUSIC, CALCULATOR, ATLAS,
    PAPER, CARD, CHIMNEY, TICKET, CODE, BIRD
580 DATA FISH, COLOUR, LAKE, TYRE, CAKE, ITCH,
    ROAD, SAIL, SAFE, PRICE, MIST, BREAD, PIE,
    INSECT
590 DATA END
600 PRINT "you guessed ";ND;"correctly.":
610 END
```

Morse Code

This program generates random letters of morse code. An optional display and range of speeds is included. At maximum speed it will generate approx. sixty characters per minute, which is adequate for the Radio Amateurs morse test.

The duration of each letter can be adjusted by changing the number in line 330.

```
100 REM
110 REM      MORSE CODE
120 REM
130 CALL CLEAR
140 RANDOMIZE
150 DIM M$(26)
160 REM      READ IN DATA
170 FOR J=1 TO 26
180 READ M$(J)
190 NEXT J
200 INPUT "SPEED (1 TO 500)? ":SP
210 PRINT : : : :
220 IF (SP>500)+(SP<1)THEN 200
230 INPUT "DISPLAY (Y/N)? ":Q$
240 PRINT : : : :
250 IF Q$(">")"Y" THEN 280
260 DISP=1
270 REM      MAIN LOOP
280 R=INT(RND*26)+1
290 IF DISP=0 THEN 310
300 PRINT CHR$(R+64);" ";
310 FOR J=1 TO LEN(M$(R))
320 CALL SOUND(50*VAL(SEG$(M$(R),J,1)),
330,2)
```

```
330 FOR DEL=1 TO 25
340 NEXT DEL
350 NEXT J
360 FOR DEL=1 TO 540-SF
370 NEXT DEL
380 GOTO 280
390 DATA 13,3111,3131,311,1,1131,331,
      1111,11,1333,313,1311,33
400 DATA 31,333,1331,3313,131,111,3,113,
      1113,133,3113,3133,3311
```

Penguin

It's no fun being a penguin nowadays. Six eggs to protect from marauding birds and only a laser cannon for company!

See how you fare as a penguin.

Move with the two keys S (left) and D (right). Fire your laser with the space bar.

The game ends when your six eggs are destroyed.

```
100 REM
110 REM   PENGUIN
120 REM
130 RANDOMIZE
140 CALL CLEAR
150 GOTO 470
160 REM   MOVE SUBROUTINE
170 CALL KEY(0,KY,S)
180 IF S=0 THEN 200
190 CALL HCHAR(23,PC,32)
200 IF KY<>83 THEN 220
210 PC=PC-1
220 IF KY<>68 THEN 240
230 PC=PC+1
240 IF PC<33 THEN 260
250 PC=32
260 IF PC>3 THEN 280
270 PC=3
280 CALL HCHAR(23,PC,130)
290 IF KY<>32 THEN 450
300 IF FRF=1 THEN 450
```



```

310 CALL SOUND(100,-5,3)
320 CALL VCHAR(2,PC,150,21)
330 CALL VCHAR(2,PC,32,21)
340 FRF=1
350 IF (PC=BC)*(BR<23)THEN 360 ELSE 450
360 CALL SOUND(400,-6,2)
370 CALL SCREEN(7)
380 GOSUB 170
390 SC=SC+100
400 CALL SCREEN(6)
410 M(WE)=0
420 GOSUB 1170
430 GOSUB 170
440 GOSUB 990
450 RETURN
460 REM  DEFINE CHARACTERS
470 CALL CHAR(131,"0000183C3C3C3C18")
480 CALL CHAR(130,"1C383C666666663C")
490 CALL CHAR(150,"0808080808080808")
500 CALL COLOR(15,16,1)
510 FOR J=1 TO 6
520 M(J)=0
530 NEXT J
540 SC=0
550 CALL CLEAR
560 CALL SCREEN(6)
570 PC=15
580 FOR J=1 TO 6
590 EC(J)=J*5
600 CALL HCHAR(24,J*5,131)
610 NEXT J
620 EL=6
630 GOSUB 1170

```

```

640 FOR J=1 TO 6
650 GOSUB 990
660 GOSUB 170
670 CALL HCHAR(BR,BC,ST)
680 IF BC<=EC(WE)THEN 710
690 BC=BC-1
700 CALL CHAR(140,"0082CEFC38786E47")
710 IF BC>=EC(WE)THEN 740
720 BC=BC+1
730 CALL CHAR(140,"0141733F1C1E76E2")
740 IF BR>23 THEN 760
750 BR=BR+1
760 CALL GCHAR(BR,BC,ST)
770 CALL HCHAR(BR,BC,140)
780 GOSUB 160
790 IF (BC=EC(WE))* (BR=24)THEN 800
    ELSE 850
800 CALL SOUND(200,110,3)
810 CALL HCHAR(BR,BC,32)
820 EL=EL-1
830 GOSUB 1170
840 GOTO 860
850 GOTO 670
860 NEXT J
870 REM  END OF GAME
880 FOR J=400 TO 120 STEP -20
890 CALL SOUND(100,J,2)
900 NEXT J
910 CALL CLEAR
920 IF SC>HS THEN 930 ELSE 940
930 HS=SC
940 PRINT "YOUR SCORE =",SC: : : :
950 PRINT "HIGH SCORE =",HS: : : : :

```

```

960 INPUT "PLAY AGAIN? ":Q$
970 IF (SEG$(Q$,1,1)="N")+ (SEG$(Q$,1,
    1)="n") THEN 980 ELSE 510
980 END
990 BC=INT(RND*28)+3
1000 BR=INT(RND*17)+2
1010 FRF=0
1020 ST=32
1030 WE=INT(RND*6)+1
1040 GOSUB 170
1050 IF M(WE)=1 THEN 1030
1060 M(WE)=1
1070 IF BC<16 THEN 1100
1080 CALL CHAR(140,"8082CEFC38786E47")
1090 GOTO 1110
1100 CALL CHAR(140,"0141733F1C1E76E2")
1110 RETURN
1120 FOR PR=1 TO LEN(M$)
1130 CALL HCHAR(1,CL+PR,ASC(SEG$(M$,
    PR,1)))
1140 NEXT PR
1150 RETURN
1160 REM PRINT SCORE
1170 M$="EGGS LEFT = "&STR$(EL)
1180 CL=3
1190 GOSUB 1120
1200 GOSUB 170
1210 M$="SCORE = "&STR$(SC)
1220 CL=18
1230 GOSUB 1120
1240 RETURN

```

Lunar Lander

You are at the controls of a lunar landing module. Control its rate of descent by pressing the space bar, but watch your fuel consumption.

```
100 REM
110 REM    LUNAR LANDER
120 REM
130 CALL CLEAR
140 CALL SCREEN(15)
150 REM    DEFINE CHARACTERS
160 CALL CHAR(149,"FFFFFFFFFFFFFFFF")
170 CALL CHAR(130,"7EFFFFFFFF7E8181")
180 CALL CHAR(140,"3C3C1C18181")
190 CALL COLOR(13,5,1)
200 CALL COLOR(14,7,1)
210 CALL COLOR(15,4,1)
220 HT=2
230 VEL=0
240 FU=50
250 REM    PLOT LUNAR SURFACE
260 CALL HCHAR(24,3,149,29)
270 CALL HCHAR(23,3,149,5)
280 CALL HCHAR(23,26,149,6)
290 CALL HCHAR(22,27,149,5)
300 CALL HCHAR(22,3,149,2)
310 REM    MAIN LOOP
320 CALL HCHAR(HT,16,130)
330 VEL=VEL+1.2
340 M$="FUEL = "&STR$(FU)
350 FOR J=1 TO LEN(M$)
360 CH=ASC(SEG$(M$,J,1))
```

```

370 CALL HCHAR(1,11+J,CH)
380 NEXT J
390 IF FU=0 THEN 510
400 REM    IGNITE?
410 CALL KEY(0,K,S)
420 IF K<>32 THEN 510
430 CALL SOUND(100,-5,2)
440 CALL GCHAR(HT+1,16,CH)
450 CALL HCHAR(HT+1,16,140)
460 FU=FU-10
470 IF FU>0 THEN 490
480 FU=0
490 VEL=VEL-2
500 CALL HCHAR(HT+1,16,CH)
510 CALL HCHAR(HT,16,32)
520 HT=INT(HT+VEL)
530 IF HT>=23 THEN 560
540 GOTO 320
550 REM    CONTACT
560 IF VEL<2.1 THEN 610
570 CALL SOUND(300,-5,2)
580 CALL HCHAR(24,16,131)
590 PRINT : "          YOU CRASHED!" :
600 GOTO 640
610 CALL SOUND(200,320,2)
620 CALL HCHAR(23,16,130)
630 PRINT : "          A PERFECT LANDING!"
640 FOR DEL=1 TO 1000
650 NEXT DEL
660 GOTO 130

```

Space Zap

A game for 2 players. Player one moves his ship using keys Q and A, and fires with key X. Player two moves with keys O and L and fires with M.

The aim is to destroy the opposing players ship. Nine hits signal the destruction of a ship. Fast fingers are needed since the normal repeat key facility of the TI 99 is not used!

```
100 REM
110 REM      SPACE ZAP
120 REM
130 GOTO 720
140 CALL SOUND(100,-5,2)
150 CALL HCHAR(VR,C1+1,150,26)
160 IF R1<>R2 THEN 180
170 GOTO 550
180 CALL HCHAR(VR,C1+1,32,26)
190 CALL KEY(0,K,S)
200 IF S=-1 THEN 360
210 IF K<>88 THEN 250
220 VR=R1
230 HIT=1
240 GOSUB 140
250 IF K<>81 THEN 270
260 R1=R1-1
270 IF K<>65 THEN 290
280 R1=R1+1
290 IF R1=R10 THEN 360
300 IF (R1>2)*(<R1<24) THEN 330
310 R1=R10
320 GOTO 360
```

```

330 CALL VCHAR(R10,C1,32)
340 CALL VCHAR(R1,C1,130)
350 R10=R1
360 RETURN
370 CALL KEY(0,K,S)
380 IF S=-1 THEN 540
390 IF K<>77 THEN 430
400 VR=R2
410 HIT=2
420 GOSUB 140
430 IF K<>79 THEN 450
440 R2=R2-1
450 IF K<>76 THEN 470
460 R2=R2+1
470 IF R2=R20 THEN 540
480 IF (R2>2)*(R2<24) THEN 510
490 R2=R20
500 GOTO 540
510 CALL VCHAR(R20,C2,32)
520 CALL VCHAR(R2,C2,140)
530 R20=R2
540 RETURN
550 REM
560 CALL SOUND(300,-5,2)
570 IF HIT=1 THEN 630
580 SC2=SC2+1
590 CALL HCHAR(1,22,ASC(STR$(SC2)))
600 IF SC2=9 THEN 680
610 HIT=0
620 GOTO 670
630 SC1=SC1+1
640 CALL HCHAR(1,11,ASC(STR$(SC1)))
650 IF SC1=9 THEN 680

```

```

660 HIT=0
670 GOTO 180
680 CALL HCHAR(VR,C1+1,32,26)
690 FOR DEL=1 TO 1500
700 NEXT DEL
710 END
720 CALL CLEAR
730 CALL SCREEN(15)
740 R10=12
750 R20=12
760 R1=12
770 R2=12
780 C1=4
790 C2=31
800 CALL CHAR(130,"F0F8FCFEFEFCF8F0")
810 CALL CHAR(140,"0F1F3F7F7F3F1F0F")
820 CALL CHAR(150,"000000FFFF")
830 CALL COLOR(15,16,1)
840 CALL VCHAR(R1,C1,130)
850 CALL VCHAR(R2,C2,140)
860 GOSUB 190
870 GOSUB 370
880 GOTO 860

```


Kaleidoscope

This program plots a series of coloured blocks against a changing background.

Although only short, it provides a fascinatingly hypnotic display of the colour range of the TI 99.

```
100 REM
110 REM    KALEIDOSCOPE
120 REM
130 CALL CLEAR
140 CALL SCREEN(15)
150 FOR J=32 TO 152 STEP 8
160 CALL CHAR(J,"FFFFFFFFFFFFFFFF")
170 CALL COLOR((J/8)-3,INT(RND*16+1),1)
180 CALL HCHAR(INT(RND*24+1),INT(RND*29
    +3),J)
190 NEXT J
200 GOTO 150
```

Chase

This game has you being chased around the screen by a chomping monster, to the accompaniment of some colourful sounds!

Guide your man around the screen using the arrow keys (E S D X), but keep your fingers moving fast to survive!

A point of interest to note is the redefining of characters in lines 720 to 780 to give an impression of movement to the two figures.

Masochists might like to try inserting an extra line:

```
735 GOSUB 510
```

```
100 REM
110 REM    CHASE
120 REM
130 CALL CLEAR
140 CALL SCREEN(8)
150 CALL CHAR(130,"3844386CAA3844C6")
160 CALL COLOR(13,7,1)
170 CALL CHAR(140,"387CBAFE82FE3838")
180 CALL COLOR(14,3,1)
190 R=23
200 RO=23
210 C=16
220 CO=16
230 CALL HCHAR(R,C,130)
240 MR=2
250 MRO=2
260 MC=16
270 MCO=16
280 GOTO 700
```

```

290 REM    MOVE MAN
300 CALL KEY(0,K,S)
310 IF K<>69 THEN 330
320 R=R-1
330 IF K<>88 THEN 350
340 R=R+1
350 IF K<>68 THEN 370
360 C=C+1
370 IF K<>83 THEN 390
380 C=C-1
390 IF (R=R0)*(C=C0)THEN 490
400 IF (R<2)+(R>23)+(C>31)+(C<3)THEN 410
    ELSE 440
410 R=R0
420 C=C0
430 GOTO 490
440 CALL SOUND(100,330,2)
450 CALL HCHAR(R0,C0,32)
460 CALL HCHAR(R,C,130)
470 R0=R
480 C0=C
490 RETURN
500 REM    MOVE MONSTER
510 IF MR>=R THEN 530
520 MR=MR+1
530 IF MR<=R THEN 550
540 MR=MR-1
550 IF MC>=C THEN 570
560 MC=MC+1
570 IF MC<=C THEN 590
580 MC=MC-1
590 CALL SOUND(100,220,2)
600 CALL HCHAR(MR0,MC0,32)

```

```

610 CALL HCHAR(MR,MC,140)
620 IF (MR=R)*(MC=C)THEN 660
630 MR0=MR
640 MC0=MC
650 RETURN
660 CALL SOUND(200,200,2)
670 PRINT "          CAUGHT YOU!":...
680 END
690 REM    MAIN LOOP
700 GOSUB 300
710 GOSUB 510
720 CALL CHAR(130,"3844386C8A38286C")
730 GOSUB 300
740 CALL CHAR(130,"3844386C8A3844C6")
750 CALL CHAR(140,"387CBAFEFEFE3838")
760 GOSUB 300
770 GOSUB 510
780 CALL CHAR(140,"387CBAFE82FE3838")
790 GOTO 700

```

Digital Clock

This program displays a working digital clock in the centre of the screen.

If the alarm function is not required, then enter any number greater than 23 for the hour when setting it.

The speed of the clock can be adjusted via the value in line 560.

```
100 REM
110 REM    DIGITAL CLOCK
120 REM
130 CALL CLEAR
140 PRINT "PLEASE ENTER THE CORRECT
    TIME":
150 INPUT "HOUR? ":HR
160 INPUT "MINUTES? ":MIN
170 PRINT ::"SET ALARM":
180 INPUT "HOUR? ":AH
190 INPUT "MINUTE? ":AM
200 CALL CLEAR
210 CALL SCREEN(16)
220 CALL CHAR(130,"FFFFFFFFFFFFFFFF")
230 CALL COLOR(13,7,1)
240 REM    DRAW BOX
250 CALL HCHAR(10,12,130,9)
260 CALL HCHAR(14,12,130,9)
270 CALL VCHAR(11,12,130,3)
280 CALL VCHAR(11,20,130,3)
290 REM    CALCULATE TIME
300 IF MIN<=59 THEN 330
310 HR=HR+1
```

```

320 MIN=0
330 IF HR<=23 THEN 350
340 HR=0
350 IF (HR<10)*(MIN<10)THEN 360 ELSE 380
360 T$=" "&STR$(HR)&":"&"0"&STR$(MIN)
370 GOTO 460
380 IF MIN>9 THEN 410
390 T$=" "&STR$(HR)&":"&"0"&STR$(MIN)&" "
400 GOTO 460
410 IF HR>9 THEN 440
420 T$=" "&STR$(HR)&":"&STR$(MIN)
430 GOTO 460
440 T$=" "&STR$(HR)&":"&STR$(MIN)
450 REM PRINT TIME
460 FOR J=1 TO LEN(T$)
470 CH=ASC(SEG$(T$,J,1))
480 CALL HCHAR(12,12+J,CH)
490 NEXT J
500 REM DELAY
510 FOR F=1 TO 60
520 CALL HCHAR(12,16,32)
530 FOR D=1 TO 50
540 NEXT D
550 CALL HCHAR(12,16,58)
560 FOR D=1 TO 241
570 NEXT D
580 IF (HR=AM)*(MIN=AM)THEN 590 ELSE 600
590 CALL SOUND(100,330,1)
600 NEXT F
610 MIN=MIN+1
620 GOTO 300

```

Binary to Hex Conversion

This program inputs a sequence of binary digits and converts them to a hexadecimal number. Programmers will appreciate its use in character definition on the TI 99.

```
100 REM
110 REM    BINARY TO HEXADECIMAL
120 REM    CONVERSION
130 REM
140 CALL CLEAR
150 PRINT "INPUT BINARY AS A STRING OF
      1'S AND ZERO'S,":
160 PRINT "E.G. 100110":
170 PRINT "INPUT S TO STOP":
180 INPUT "BINARY? ":B$
190 IF B$="S" THEN 420
200 IF LEN(B$)/4=INT(LEN(B$)/4)THEN 230
210 B$="0"&B$
220 GOTO 200
230 FOR J=4 TO LEN(B$)STEP 4
240 A$=SEG$(B$,LEN(B$)-J+1,4)
250 N=1
260 HEX=""
270 FOR K=1 TO 4
280 H$=SEG$(A$,5-K,1)
290 IF H$="0" THEN 310
300 HEX=HEX+N
310 N=N*2
320 NEXT K
330 IF HEX<10 THEN 360
```

```
340 HE#=CHR$(HEX+55)
350 GOTO 370
360 HE#=STR$(HEX)
370 HV#=HE#&HV#
380 NEXT J
390 PRINT "HEXADECIMAL = ",HV#
400 HV#=""
410 GOTO 180
420 END
```


Hex to Decimal Conversion

Inputs a valid hexadecimal number and transforms it to decimal. This program will be of particular interest to those studying machine code.

```
100 REM
110 REM    HEXADECIMAL TO DECIMAL
120 REM    CONVERSION
130 REM
140 CALL CLEAR
150 PRINT "INPUT HEXADECIMAL NUMBERS ASA
    SINGLE STRING,"::
160 PRINT "E.G.  154EA"::
170 PRINT "INPUT S TO STOP"::
180 INPUT "HEXADECIMAL? " :H$
190 IF H$="S" THEN 330
200 N=1
210 DEC=0
220 FOR J=1 TO LEN(H$)
230 A$=SEG$(H$,LEN(H$)-J+1,1)
240 IF ASC(A$)>58 THEN 270
250 HT=VAL(A$)
260 GOTO 280
270 HT=ASC(A$)-55
280 DEC=DEC+N*HT
290 N=N*16
300 NEXT J
310 PRINT "DECIMAL =",DEC::
320 GOTO 180
330 END
```

Binary to Decimal Conversion

This program will convert a sequence of binary digits (0's and 1's) into a decimal number. The length of the binary number is limited only by the maximum length of the input statement on the TI 99!

```
100 REM
110 REM    BINARY TO DECIMAL
120 REM    CONVERSION
130 REM
140 CALL CLEAR
150 CALL SCREEN(16)
160 PRINT "INPUT YOUR BINARY NUMBERS ASA
      STRING OF 1'S & ZERO'S,":
170 PRINT "E.G.  100110":
180 PRINT "INPUT S TO STOP":
190 REM    MAIN LOOP
200 INPUT "BINARY? " : B$
210 REM    TEST FOR END
220 IF B$="S" THEN 340
230 N=1
240 DEC=0
250 FOR J=1 TO LEN(B$)
260 A$=SEG$(B$,LEN(B$)-J+1,1)
270 IF A$="0" THEN 290
280 DEC=DEC+N
290 N=N*2
300 NEXT J
310 REM    PRINT DECIMAL
320 PRINT "DECIMAL =",DEC:
```

330 GOTO 200
340 END

Arithmetic

An arithmetic test for all ages.

The inclusion of a maximum number makes this program suitable for use at all levels.

It will generate an endless sequence of addition and subtraction sums, testing for the correct answer in each case.

To stop the program, press the function and clear keys simultaneously.

```
100 REM
110 REM    ARITHMETIC
120 REM
130 RANDOMIZE
140 CALL CLEAR
150 CALL SCREEN(16)
160 CALL CHAR(130,"00FFFF")
170 PRINT "ARITHMETIC TEST":...:
180 INPUT "ENTER MAXIMUM NUMBER ":MN
190 IF MN<1 THEN 180
200 C=14
210 CALL CLEAR
220 REM    SELECT SIGN
230 R=INT(RND*2)
240 IF R=1 THEN 270
250 R=-1
260 REM    SELECT NUMBERS
270 N1=INT(RND*MN)+1
280 N2=INT(RND*MN)
290 IF N1<N2 THEN 270
300 REM    PRINT NUMBERS
```

```

310 M$=STR$(N1)
320 P=12
330 GOSUB 780
340 M$=STR$(N2)
350 IF LEN(M$)=LEN(STR$(N1)) THEN 380
360 M$=" "&M$
370 GOTO 350
380 P=13
390 GOSUB 780
400 IF R<1 THEN 430
410 CALL HCHAR(12,C+LEN(STR$(N1))+1,43)
420 GOTO 440
430 CALL HCHAR(12,C+LEN(STR$(N1))+1,45)
440 CALL HCHAR(14,C+1,130,LEN(M$))
450 REM INPUT ANSWER
460 FOR J=1 TO LEN(STR$(N1+N2*R))
470 CALL KEY(0,K,S)
480 IF S=0 THEN 470
490 CALL SOUND(50,290,4)
500 IF (K<48)+(K>57) THEN 470
510 CALL HCHAR(15,C+LEN(STR$(N1))+1-J,K)
520 AN$=CHR$(K)&AN$
530 NEXT J
540 IF VAL(AN$)=N1+N2*R THEN 680
550 REM WRONG ANSWER
560 M$="TRY AGAIN!"
570 OF=2
580 P=22
590 CALL SOUND(150,120,2)
600 GOSUB 780
610 OF=0
620 GOSUB 830
630 CALL HCHAR(22,C-4,32,13)

```

```
640 CALL HCHAR(15,C-1,32,14)
650 AN$=""
660 GOTO 460
670 REM    RIGHT ANSWER
680 M$="CORRECT!"
690 OF=2
700 P=22
710 CALL SOUND(150,330,2)
720 GOSUB 780
730 OF=0
740 GOSUB 830
750 CALL CLEAR
760 AN$=""
770 GOTO 230
780 FOR N=1 TO LEN(M$)
790 CH=ASC(SEG$(M$,N,1))
800 CALL HCHAR(P,C+N-OF,CH)
810 NEXT N
820 RETURN
830 FOR DE=1 TO 1000
840 NEXT DE
850 RETURN
```

Spelling Test

This program prints out a list of three words of different spelling, only one of which is correct. If the correct spelling is entered, a new word is chosen.

The list of words in the data statements can, of course, be varied to suit any age. A variation would be to make none of the spellings correct!

```
100 REM
110 REM   SPELLING TEST
120 REM
130 CALL CLEAR
140 CALL SCREEN(8)
150 RANDOMIZE
160 DIM WD$(20,3)
170 DEF RX=INT(RND*3)+1
180 PRINT "SPELLING TEST":...:
190 REM   READ IN DATA
200 FOR J=1 TO 20
210 FOR K=1 TO 3
220 READ WD$(J,K)
230 NEXT K
240 NEXT J
250 REM   CHOOSE WORD & ORDER
260 R1=RX
270 R2=RX
280 IF R1=R2 THEN 270
290 R3=RX
300 IF (R3=R1)+(R3=R2) THEN 290
310 RD=INT(RND*20)+1
320 IF RD=RWD THEN 310
```

```

330 RWD=RD
340 PRINT "1. ";WD$(RWD,R1)
350 PRINT "2. ";WD$(RWD,R2)
360 PRINT "3. ";WD$(RWD,R3): : :
370 PRINT "WHAT IS THE CORRECT SPELLING";
380 INPUT AN$
390 PRINT : :
400 IF AN$=WD$(RWD,1)THEN 430
410 PRINT "TRY AGAIN!": :
420 GOTO 380
430 PRINT "CORRECT!": :
440 GOTO 260
450 DATA CARROT,CAROT,CAROTT
460 DATA TELEVISION,TELIVISION,TELEVISON
470 DATA COMPUTER,COMPUTOR,COMPUTAR
480 DATA VEHICLE,VEICLE,VEERCLE
490 DATA RANDOM,RANDEM,RANDAM
500 DATA FEBRUARY,FEBUARY,FEBURARY
510 DATA YELLOW,YELLOW,YELLEW
520 DATA SENTENCE,SENTANCE,SENTENSE
530 DATA SUCCESS,SUCCESS,SUCCESE
540 DATA TOMORROW,TOMMOROW,TOMOROW
550 DATA ENVIRONMENT,ENVIROMENT,ENVIRAMENT
560 DATA BEAUTIFUL,BUEATIFUL,BEAUTIFULL
570 DATA IMPOSSIBLE,IMPOSIBLE,INPOSSIBLE
580 DATA ILLEGIBLE,ILEGIBLE,ILLEGABLE
590 DATA AMEND,AMMEND,EMEND
600 DATA ORCHESTRA,ORKESTRA,ORCESTRA
610 DATA INTRIGUE,INTRIGE,INTREAGUE
620 DATA PNEUMATIC,NEWMATIC,PNUMATIC
630 DATA WEDNESDAY,WENSDAY,WENDSDAY
640 DATA APPLICATION,APLICATION,
    APPLICATION

```


Multiplication Tables

Here's a simple program to test your knowledge of multiplication. The range of numbers is controlled by a simple input statement at the beginning, which makes the program suitable for a range of ages.

```
100 REM
110 REM      MULTIPLICATION TABLES
120 REM
130 CALL CLEAR
140 RANDOMIZE
150 INPUT "MAXIMUM NUMBER? ":MAX
160 N1=INT(RND*MAX)+1
170 N2=INT(RND*MAX)+1
180 PRINT : "WHAT IS";N1;"TIMES";N2;
190 INPUT ANS
200 IF ANS=N1*N2 THEN 230
210 PRINT : "TRY AGAIN!":
220 GOTO 190
230 PRINT : "CORRECT!":
240 INPUT "AGAIN? ":Q$
250 IF Q$="N" THEN 270
260 GOTO 160
270 END
```

Report

This program produces a list of useful sounding phrases for incorporation into reports or speeches. Readers might like to insert their own words into the data statements in lines 240–300!

A similar structure is used in many of the “poetry” generating programs around. Just insert your own phrases instead of the single words.

```
100 REM
110 REM   REPORT
120 REM
130 RANDOMIZE
140 CALL CLEAR
150 REM   READ IN DATA
160 FOR J=1 TO 7
170 READ A$(J),B$(J),C$(J)
180 NEXT J
190 INPUT "HOW MANY PHRASES? ":PH
200 PRINT : : : :
210 REM   PRINT PH PHRASES
220 FOR J=1 TO PH
230 PRINT A$(INT(RND*7)+1);" ";
240 PRINT B$(INT(RND*7)+1);" ";
250 PRINT C$(INT(RND*7)+1): : : :
260 NEXT J
270 DATA ON-GOING,HUMANISTIC,ENVIRONMENT
280 DATA MODULAR,INTEGRATED,SITUATION
290 DATA DIFFERENTIATED,MOTIVATIONAL,
    PROCESS
300 DATA MINORITY,CREATIVE,EXPERIMENT
310 DATA OPEN-ENDED,VERTICAL,TECHNIQUE
```

320 DATA INDIVIDUAL,EVALUATIVE,RESOURCE
330 DATA HOMOJENOUS,OBJECTIVE,FACILITY

Planet Lander

This is a computer simulation of a spacecraft landing on any one of six planets. Purists will no doubt be aware that the moon is not a planet!

Full instructions are included in the listing, so are not reported here.

```
100 REM
110 REM    PLANET LANDER
120 REM
130 CALL CLEAR
140 CALL SCREEN(16)
150 REM    INITIALISE VARIABLES
160 T=0
170 HT=500
180 FRF=0
190 V=20
200 DEF TRN(X)=INT(X*10)/10
210 REM    PRINT INSTRUCTIONS
220 PRINT "PLANET LANDER":
230 PRINT "this is a computer simulation
      of a space craft landing on any one
      of six    Planets.":
240 PRINT "use lower case letters for
      your reply.":
250 INPUT "name your Planet (mercury/
      venus/earth/the moon/mars/
      Pluto) ":Q$
260 PRINT :
270 REM    TEST FOR PLANET
280 IF Q$(">"mercury" THEN 320
```

```

290 G=3.76
300 FR=500
310 GOTO 520
320 IF Q$(">"venus" THEN 360
330 G=8.77
340 FR=950
350 GOTO 520
360 IF Q$(">"earth" THEN 400
370 G=9.81
380 FR=1000
390 GOTO 520
400 IF Q$(">"the moon" THEN 440
410 G=1.62
420 FR=250
430 GOTO 520
440 IF Q$(">"mars" THEN 480
450 G=3.8
460 FR=500
470 GOTO 520
480 IF Q$(">"pluto" THEN 250
490 G=4
500 FR=500
510 REM MORE INSTRUCTIONS
520 PRINT "you are ";HT;"metres";"above
the surface of ";Q$;"."::
530 PRINT "your onboard ZX81 will not
load your auto-lander Program, so you
will have to input instructions
direct";
540 PRINT "to the texas master computer
by hand."::
550 PRINT "to further compound your
problems you are desperately short of
fuel!"::

```

```

560 PRINT "when a question mark appears,
      type % engine burn required and
      press enter."::
570 PRINT "time height velocity fuel"::
580 PRINT T;TAB(6);TRN(HT);TAB(14);
      TRN(V);TAB(22);INT(FR)::
590 IF FR<>1 THEN 620
600 BR=0
610 GOTO 720
620 INPUT BR
630 PRINT ::
640 IF (BR>100)+(BR<0)THEN 620
650 IF BR<FR THEN 710
660 BR=FR
670 FRF=1
680 PRINT "out of fuel"::
690 FR=0
700 GOTO 720
710 FR=FR-BR
720 REM    NEW HEIGHT
730 HT=HT-V
740 IF HT<=0 THEN 790
750 V=V+G-BR/5
760 T=T+1
770 GOTO 570
780 REM    LANDED
790 IF V<1 THEN 840
800 IF V<4 THEN 870
810 CALL SOUND(1000,-6,2)
820 PRINT "c r a s h !!!"::
830 GOTO 890
840 CALL SOUND(250,262,2)
850 PRINT "a Perfect landing!"::
860 GOTO 890

```

```
870 CALL SOUND(1000,-6,2)
880 PRINT "landed with craft damage -
      await rescue!":
890 INPUT "another game?":Q$
900 IF (SEG$(Q$,1,1)="y")+ (SEG$(Q$,1,1)
      ="Y")THEN 130
910 END
```

Entrapment

A game of strategy for 2 players.

Each player controls a different coloured snake which winds its way around the screen.

Player one uses keys E S D X and player two uses IJKM to direct the movement of their snake.

A player loses if his snake hits any obstacle, including its own body, or tries to leave the screen!

```
100 REM
110 REM    ENTRAPMENT
120 REM
130 GOTO 640
140 REM    PLAYER ONE
150 CALL KEY(0,KY,S)
160 IF (KY=69)+(KY=88)+(KY=83)+(KY=68)
    THEN 170 ELSE 180
170 K1=KY
180 IF (K1<>69)+(R1<=1)THEN 200
190 R1=R1-1
200 IF (K1<>88)+(R1>=24)THEN 220
210 R1=R1+1
220 IF (K1<>83)+(C1<=3)THEN 240
230 C1=C1-1
240 IF (K1<>68)+(C1>=32)THEN 260
250 C1=C1+1
260 CALL GCHAR(R1,C1,X)
270 IF X<>32 THEN 490
280 CALL SOUND(100,290,2)
```



```

290 CALL HCHAR(R1,C1,130)
300 RETURN
310 REM    PLAYER TWO
320 CALL KEY(0,KY,S)
330 IF (KY=73)+(KY=77)+(KY=74)+(KY=75)
    THEN 340 ELSE 350
340 K2=KY
350 IF (K2<>73)+(R2<=1)THEN 370
360 R2=R2-1
370 IF (K2<>77)+(R2>=24)THEN 390
380 R2=R2+1
390 IF (K2<>74)+(C2<=3)THEN 410
400 C2=C2-1
410 IF (K2<>75)+(C2>=32)THEN 430
420 C2=C2+1
430 CALL GCHAR(R2,C2,X)
440 IF X<>32 THEN 560
450 CALL SOUND(100,390,2)
460 CALL HCHAR(R2,C2,140)
470 RETURN
480 REM    END OF GAME
490 CALL SOUND(300,-5,2)
500 FOR J=1 TO 20
510 CALL HCHAR(R1,C1,32)
520 CALL SOUND(50,220,2)
530 CALL HCHAR(R1,C1,130)
540 NEXT J
550 GOTO 100
560 CALL SOUND(300,-6,2)
570 FOR J=1 TO 20
580 CALL HCHAR(R2,C2,32)
590 CALL SOUND(50,280,2)
600 CALL HCHAR(R2,C2,140)

```

```
610 NEXT J
620 GOTO 100
630 REM   MAIN PROGRAM
640 CALL CLEAR
650 CALL SCREEN(8)
660 CALL CHAR(130,"183C7EFFFF7E3C18")
670 CALL CHAR(140,"183C7EFFFF7E3C18")
680 CALL COLOR(13,7,1)
690 CALL COLOR(14,6,1)
700 R1=2
710 K1=68
720 C1=4
730 R2=23
740 K2=74
750 C2=31
760 GOSUB 150
770 GOSUB 320
780 GOTO 760
```

Print At

This subroutine simulates the DISPLAY AT or PRINT AT statement which finds extensive use in other dialects of, or in extended, Basic.

To use this, the print line is placed in a variable P\$ and the screen line and column number in X and Y. It can then be accessed as a normal subroutine.

```
100 REM
110 REM      PRINT AT X,Y,M$
120 REM      ROUTINE
130 REM
140 FOR J=1 TO LEN(P$)
150 IF Y<32 THEN 180
160 Y=3
170 X=X+1
180 IF X<24 THEN 200
190 X=1
200 CH=ASC(SEG$(P$,J,1))
210 CALL HCHAR(X,Y,CH)
220 Y=Y+1
230 NEXT J
```

Cheque Book

A cheque book balancer to help you keep track of your account. To use, enter your last known balance followed by a list of cheques written out and payments in. Each list is ended with the input of a zero. The computer will prompt you for each list.

```
100 REM
110 REM    CHEQUE BOOK
120 REM
130 CALL CLEAR
140 INPUT "YOUR LAST BALANCE? " : BAL
150 PRINT : : : "CHEQUES SINCE:" : :
160 INPUT CHQ
170 IF CHQ=0 THEN 200
180 BAL=BAL-CHQ
190 GOTO 160
200 PRINT : : : "PAYMENTS IN:" : :
210 INPUT PAY
220 IF PAY=0 THEN 250
230 BAL=BAL+PAY
240 GOTO 210
250 PRINT : : : "YOUR CURRENT BALANCE
    IS" : BAL
260 END
```

Metric Converter

This is a useful demonstration program to convert metric length or weight to Imperial. Further routines could be easily added to cover different conversions.

```
100 REM
110 REM    METRIC CONVERTER
120 REM
130 CALL CLEAR
140 PRINT "1. KILOGRAMS TO POUNDS":
150 PRINT "2. METRES TO FEET & INS":
160 INPUT "INPUT 1 OR 2 TO SELECT " KY
170 IF (KY=1)+(KY=2) THEN 180 ELSE 160
180 ON KY GOTO 200,320
190 REM    KILOS TO LBS
200 PRINT : "INPUT METRIC:"
210 INPUT KGS
220 IF KGS=0 THEN 430
230 OZS=KGS*2.2046*16
240 IF INT(OZS)<16 THEN 270
250 PDS=INT(OZS/16)
260 OZS=OZS-PDS*16
270 PRINT PDS;"LBS";OZS;"OZ"
280 PDS=0
290 OZS=0
300 GOTO 210
310 REM    METRES TO FT & INS
320 PRINT : "INPUT METRIC:"
330 INPUT MTS
340 IF MTS=0 THEN 430
350 INS=MTS*39.37
360 IF INT(INS)<12 THEN 390
```

```
370 FT=INT(INS/12)
380 INS=INS-FT*12
390 PRINT FT;"FT";INS;"IN":
400 FT=0
410 INS=0
420 GOTO 330
430 END
```

Alien Attack

You are being attacked by a swarm of aliens coming at you from the bottom of the screen.

Guide your space craft using the keys S and D. Fire with the space bar.

The smaller green aliens score 10 points and the larger red score 25 points.

The game ends when you collide with an alien or allow 100 to escape.

```
100 REM
110 REM   ALIEN ATTACK
120 REM
130 GOSUB 380
140 GOTO 490
150 CALL GCHAR(LV,S,X)
160 IF (X=130)+(X=140)THEN 670
170 CALL HCHAR(LV,S,152)
180 FOR J=1 TO 8
190 IF HIT=1 THEN 350
200 CALL GCHAR(LV+J,S,X)
210 IF X<>130 THEN 290
220 SC=SC+10
230 NG=NG-1
240 HIT=1
250 CALL SOUND(300,-5,3)
260 CALL HCHAR(LV+J,S,141)
270 CALL HCHAR(LV+J,S,32)
280 GOTO 320
```

```

290 IF X<>140 THEN 320
300 SC=SC+25
310 GOTO 230
320 CALL VCHAR(LV+J,S,139)
330 CALL SOUND(50,-6,4)
340 CALL VCHAR(LV+J,S,32)
350 NEXT J
360 HIT=0
370 RETURN
380 CALL SCREEN(16)
390 RANDOMIZE
400 REM   DEFINE CHARACTERS
410 CALL CHAR(130,"38D64438EE9292")
420 CALL CHAR(139,"0010101010101000")
430 CALL CHAR(140,"C3FFDBFFC3BD1866")
440 CALL CHAR(152,"FFFFFFFF7E7E3C3C")
450 CALL COLOR(13,3,1)
460 CALL COLOR(14,7,1)
470 CALL COLOR(16,6,1)
480 RETURN
490 S=16
500 LV=6
510 CALL CLEAR
520 REM   PLACE ALIEN
530 IF RND>.5 THEN 600
540 NG=NG+1
550 IF NG>110 THEN 870
560 IF RND>.90 THEN 570 ELSE 590
570 CALL HCHAR(24,INT(RND*31)+1,140)
580 GOTO 600
590 CALL HCHAR(24,INT(RND*31)+1,130)
600 PRINT " "
610 CALL HCHAR(LV-2,S,32)

```



```

620 REM    MOVE OR FIRE
630 CALL KEY(0,K,ST)
640 IF ST=0 THEN 740
650 IF K<>32 THEN 670
660 GOSUB 150
670 IF K<>68 THEN 710
680 S=S+1
690 IF S>31 THEN 720
700 GOTO 740
710 IF K<>83 THEN 740
720 S=S-1
730 IF S<3 THEN 680
740 CALL GCHAR(LV,S,X)
750 IF (X=130)+(X=140) THEN 790
760 CALL HCHAR(LV,S,152)
770 GOTO 530
780 REM    HIT
790 CALL SCREEN(7)
800 CALL SOUND(500,-5,2)
810 CALL HCHAR(LV,S,141)
820 CALL COLOR(16,7,1)
830 FOR DEL=1 TO 30
840 NEXT DEL
850 CALL HCHAR(LV,S,152)
860 CALL SCREEN(16)
870 FOR J=660 TO 120 STEP -20
880 CALL SOUND(100,J,2)
890 NEXT J
900 IF SC<HSC THEN 920
910 HSC=SC
920 PRINT "SCORE =";SC:
930 PRINT "HIGH SCORE =";HSC:
940 INPUT "PLAY AGAIN? ":Q$

```

```
950 IF SEG$(Q$,1,1)="N" THEN 1000
960 SC=0
970 NG=0
980 CALL COLOR(16,6,1)
990 GOTO 140
1000 END
```

Capitals

This program will ask you for the capital of a country. Enter the right answer (spelt correctly!) and a further question will be asked. If an incorrect answer is given, you will have three more tries to get the right answer, after which the answer will be printed.

The list of countries and capitals can be extended by adding extra data statements after line 550 and increasing the numbers in lines 150, 170 and 220 to match the number of data statements.

```
100 REM
110 REM   CAPITALS
120 REM
130 CALL CLEAR
140 RANDOMIZE
150 DIM C$(20),P$(20)
160 REM   READ IN DATA
170 FOR J=1 TO 20
180 READ C$(J),P$(J)
190 NEXT J
200 CN=0
210 REM   PICK COUNTRY
220 RN=INT(RND*20)+1
230 IF RN=R THEN 220
240 R=RN
250 PRINT "WHAT IS THE CAPITAL OF ";C$(R);
260 INPUT AN$
270 IF AN$=P$(R) THEN 320
280 CN=CN+1
290 IF CN=4 THEN 340
300 PRINT ::"TRY AGAIN!"::
```

```
310 GOTO 260
320 PRINT :;"CORRECT!":;
330 GOTO 200
340 PRINT :;"THE CAPITAL OF ";C$(R);"
    IS ";P$(R);".";:;
350 GOTO 200
360 DATA AFGHANISTAN,KABUL
370 DATA ARGENTINA,BUENOS AIRES
380 DATA BELGIUM,BRUSSELS
390 DATA BRAZIL,RIO DE JANEIRO
400 DATA CANADA,OTTAWA
410 DATA CUBA,HAVANA
420 DATA DENMARK,COPENHAGEN
430 DATA EGYPT,CAIRO
440 DATA ENGLAND,LONDON
450 DATA FRANCE,PARIS
460 DATA GREECE,ATHENS
470 DATA ITALY,ROME
480 DATA JAMAICA,KINGSTON
490 DATA NIGERIA,LAGOS
500 DATA NORWAY,OSLO
510 DATA PORTUGAL,LISBON
520 DATA RUSSIA,MOSCOW
530 DATA SCOTLAND,EDINBURGH
540 DATA SPAIN,MADRID
550 DATA U.S.A.,WASHINGTON
```

Shoebox

Shoebox offers a complete home, tape-based, filing system.

It will allow you to create and update files, e.g. list of names and telephone numbers, birthdays and dates, names and addresses, etc., as well as accessing those files to search for a particular entry or to list the entire file.

By using a menu-driven modular system of programming, it is possible to add extra routines to fulfil additional functions, e.g. sort, information search, list to printer, etc.

```
100 REM
110 REM    SHOEBOX
120 REM
130 DIM R$(200), I$(200)
140 J=1
150 CALL CLEAR
160 REM    MENU
170 PRINT "1. INPUT FILE":
180 PRINT "2. UPDATE FILE":
190 PRINT "3. PRINT RECORDS":
200 PRINT "4. SEARCH RECORDS":
210 PRINT "5. WRITE FILE":
220 PRINT "6. END":
230 INPUT "WHICH NUMBER? ":CH
240 PRINT :
250 IF (CH<1)+(CH>6)THEN 230
260 ON CH GOTO 280,360,470,550,690,750
270 REM    INPUT FILE
280 OPEN #2:"CS1",INTERNAL,INPUT ,
    FIXED 128
```

```

290 INPUT #2:R$(J),I$(J)
300 IF R$(J)="END" THEN 330
310 J=J+1
320 GOTO 290
330 CLOSE #2
340 GOTO 150
350 REM    UPDATE FILE
360 PRINT "ENTER RECORDS AND INFORMATION
    SEPARATELY."::
370 PRINT "TYPE ""END"" TO FINISH."::::
380 PRINT "RECORD"J:
390 INPUT R$(J)
400 IF R$(J)="END" THEN 450
410 PRINT "INFORMATION":
420 INPUT I$(J)
430 J=J+1
440 GOTO 380
450 GOTO 150
460 REM    PRINT RECORDS
470 FOR M=1 TO J
480 PRINT R$(M)::I$(M):
490 REM    PAUSE
500 FOR DEL=1 TO 500
510 NEXT DEL
520 NEXT M
530 GOTO 150
540 REM    SEARCH RECORDS
550 INPUT "RECORD REQUIRED? ":Q$
560 PRINT :
570 FOR M=1 TO J
580 IF R$(M)=Q$ THEN 620
590 NEXT M
600 PRINT ::"RECORD NOT FOUND":

```

```
610 GOTO 650
620 PRINT R$(M)
630 PRINT I$(M)
640 REM    PAUSE
650 FOR DEL=1 TO 3000
660 NEXT DEL
670 GOTO 150
680 REM    WRITE FILE
690 OPEN #3:"CS1",INTERNAL,OUTPUT,
    FIXED 128
700 FOR M=1 TO J
710 PRINT #3:R$(M),I$(M)
720 NEXT M
730 CLOSE #3
740 GOTO 150
750 END
```

Moneybags

It's pennies from heaven, or at least dollar signs from the top of the screen, in this amusing game. You control a catcher at the bottom of the screen, moving with the S and D keys. Catch as many of the dollars as you can, but miss 20 and the game ends.

```
100 REM
110 REM    MONEYBAGS
120 REM
130 RANDOMIZE
140 CALL CLEAR
150 REM    INITIAL VALUES
160 CR=23
170 CC=15
180 CALL CHAR(130,"000000008181FFFF")
190 FOR M=1 TO 3
200 R(M)=1
210 C(M)=INT(RND*29)+3
220 NEXT M
230 GOTO 580
240 REM    MOVE DOLLARS
250 J=INT(RND*3)+1
260 CALL HCHAR(R(J),C(J),32)
270 R(J)=R(J)+1
280 IF R(J)<>23 THEN 330
290 IF C(J)<>CC THEN 330
300 CALL SOUND(100,330,2)
310 SC=SC+10
320 GOTO 370
330 IF R(J)<>24 THEN 400
340 CALL SOUND(100,220,2)
350 CN=CN+1
```



```

360 IF CN=20 THEN 620
370 C(J)=INT(RND*29)+3
380 R(J)=1
390 GOTO 420
400 CALL HCHAR(R(J),C(J),36)
410 CALL SOUND(50,220*J,3)
420 RETURN
430 REM    MOVE CATCHER
440 CALL KEY(0,K,S)
450 IF S=0 THEN 550
460 CALL HCHAR(CR,CC,32)
470 IF K<>68 THEN 490
480 CC=CC+1
490 IF K<>83 THEN 510
500 CC=CC-1
510 IF CC>3 THEN 530
520 CC=3
530 IF CC<32 THEN 550
540 CC=32
550 CALL HCHAR(CR,CC,130)
560 RETURN
570 REM    MAIN LOOP
580 GOSUB 250
590 GOSUB 440
600 GOTO 580
610 REM    GAME OVER
620 PRINT : : : : "YOUR SCORE =";SC
630 END

```

If you have enjoyed programming this selection of games and educational programs you will be interested in several adventure games which are available through:

**APEX TRADING LTD,
115 Crescent Drive South,
Brighton BN2 6SB**

Telephone Brighton (0273) 36894 for details.

GETTING STARTED with the **TEXAS TI-99/4A**

STEPHEN SHAW

Aimed at first time users,
this book takes you from setting up your machine
and guides you step by step
until you become sufficiently expert
to write your own programs.

This 'essential' book
will help you
use TI Basic
understand Extended Basic
design programs
file data on cassette

Example programs are shown in the book
along with chapters on how to increase
the capabilities of your machine
through modules, printers, RAM packs
and disk systems.

THE AUTHOR

Stephen Shaw is a recognised authority
on the TI99/4A
and contributes regularly to the *TI User*,
Home Computing Weekly and *Computer & Video Games*.

**PHOENIX
PUBLISHING ASSOCIATES**

£5.95

**AVAILABLE FROM GOOD BOOKSHOPS
OR DIRECT FROM
PHOENIX PUBLISHING ASSOCIATES,
14, VERNON ROAD,
BUSHEY, HERTS.**

price £5.95 plus 50p p/p.

AT LAST – a book full of programs for your **TEXAS TI 99/4A.**

Written for
the home user
these programs are both
fun and educational.

Now
you can
enter a 3D maze,
run a horse race,
play tennis with a friend
and even
help a penguin to save its eggs!!

You can
test your skills with anagrams,
check up on your metric conversions,
run your own filing system,
and
look after your own home accounts.

ISBN 0 9465 7600 9
PHOENIX PUBLISHING ASSOCIATES

£5.95